Lesson no. 39-43

Introduction

Till now we have been discussing the 8088 architecture which was a 16bit processor. Newer processors of the Intel series provide 32bit architecture. Till now we were in real mode of a newer processor which is basically a compatibility mode making the newer processor just a faster version of the original 8088. Switching processor in the newer 32bit mode is a very easy task. Just turn on the least significant bit of a new register called CR0 (Control Register 0) and the processor switches into 32bit mode called protected mode. However manipulations in the protected mode are very different from those in the read mode.

All registers in 386 have been extended to 32bits. The new names are EAX, EBX, ECX, EDX, ESI, EDI, ESP, EBP, EIP, and EFLAGS. The original names refer to the lower 16bits of these registers. A 32bit address register can access upto 4GB of memory so memory access has increased a lot.

As regards segment registers the scheme is not so simple. First of all we call them segment selectors instead of segment registers and they are still 16bits wide. We are also given two other segment selectors FS and GS for no specific purpose just like ES. 

The working of segment registers as being multiplied by 10 and added into the offset for obtaining the physical address is totally changed. Now the selector is just an index into an array of segment descriptors where each descriptor describes the base, limit, and attributes of a segment. Role of selector is to select on descriptor from the table of descriptors and the role of descriptor is to define the actual base address. This decouples the selection and actual definition which is needed in certain protection mechanisms introduced into this processor. For example an operating system can define the possible descriptors for a program and the program is bound to select one of them and nothing else. This sentence also hints that the processor has some sense of programs that can or cannot do certain things like change this table of descriptors. This is called the privilege level of the program and varies for 0 (highest privilege) to 3 (lowest privilege). The format of a selector is shown below.

The table index (TI) is set to 0 to access the global table of descriptors called the GDT (Global Descriptor Table). It is set to 1 to access another table, the local descriptor table (LDT) that we will not be using. RPL is the requested privilege level that ranges from 0-3 and informs what privilege level the program wants when using this descriptor. The 13bit index is the actual index into the GDT to select the appropriate descriptor. 13 bits mean that a maximum of 8192 descriptors are possible in the GDT.

The GDT itself is an array of descriptors where each descriptor is an 8byte entry. The base and limit of GDT is stored in a 48bit register called the GDTR. This register is loaded with a special instruction LGDT and is given a memory address from where the 48bits are fetched. The first entry of the GDT must always be zero. It is called the null descriptor. After that any number of entries upto a maximum of 8191 can follow. The format of a code and data descriptor is shown below. 

The 32bit base in both descriptors is scattered into different places because of compatibility reasons. The limit is stored in 20 bits but the G bit defines that the limit is in terms of bytes of 4K pages therefore a maximum of 4GB size is possible. The P bit must be set to signal that this segment is present in memory. DPL is the descriptor privilege level again related to the protection levels in 386. D bit defines that this segment is to execute code is 16bit mode or 32bit mode. C is conforming bit that we will not be using. R signals that the segment is readable. A bit is automatically set whenever the segment is accessed. The combination of S (system) and X (executable) tell that the descriptors is a code or a data descriptor. B (big) bit tells that if this data segment is used as stack SP is used or ESP is used.

Our first example is a very rudimentary one that just goes into protected mode and prints an A on the screen by directly accessing 000B8000. 

	[org 0x0100]

              jmp  start

gdt:          dd   0x00000000, 0x00000000   ; null descriptor

              dd   0x0000FFFF, 0x00CF9A00   ; 32bit code

              ;      \--/\--/    \/||||\/

              ;       |   |      | ||||+--- Base (16..23)=0 fill later
              ;       |   |      | |||+--- X=1 C=0 R=1 A=0

              ;       |   |      | ||+--- P=1 DPL=00 S=1

              ;       |   |      | |+--- Limit (16..19) = F

              ;       |   |      | +--- G=1 D=1 r=0 AVL=0

              ;       |   |      +--- Base (24..31) = 0

              ;       |   +--- Limit (0..15) = FFFF

              ;       +--- Base (0..15)=0 fill later
              dd   0x0000FFFF, 0x00CF9200   ; data

              ;      \--/\--/    \/||||\/

              ;       |   |      | ||||+--- Base (16..23) = 0

              ;       |   |      | |||+--- X=0 E=0 W=1 A=0

              ;       |   |      | ||+--- P=1 DPL=00 S=1

              ;       |   |      | |+--- Limit (16..19) = F

              ;       |   |      | +--- G=1 B=1 r=0 AVL=0

              ;       |   |      +--- Base (24..31) = 0

              ;       |   +--- Limit (0..15) = FFFF

              ;       +--- Base (0..15) = 0

gdtreg:       dw   0x17      ; 16bit limit

              dd   0         ; 32bit base (filled later)

stack:        times 256 dd 0 ; for use in p-mode

stacktop:

start:        mov  ax, 0x2401

              int  0x15           ; enable A20

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              mov  [gdt+0x08+2], ax

              shr  eax, 16

              mov  [gdt+0x08+4], al         ; fill base of code desc

              xor  edx, edx

              mov  dx, cs

              shl  edx, 4

              add  edx, stacktop            ; edx = stack top for p-mode

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              add  eax, gdt

              mov  [gdtreg+2], eax          ; fill phy base of gdt 

              lgdt [gdtreg]                 ; load gdtr

              mov  eax, cr0

              or   eax, 1

              cli                           ; MUST disable interrupts

              mov  cr0, eax                 ; P-MODE ON

              jmp  0x08:pstart              ; load cs

;;;;; 32bit protected mode ;;;;;

[bits 32] ; ask assembler to generate 32bit code

pstart:       mov  eax, 0x10

              mov  ds, ax

              mov  es, ax                   ; load other seg regs
              mov  fs, ax                   ; flat memory model

              mov  gs, ax

              mov  ss, ax

              mov  esp, edx

              mov  byte [0x000b8000], 'A'   ; direct poke at video 

              jmp  $                        ; hang around


Gate A20 is a workaround for a bug that is not detailed here. The BIOS call will simply enable it to open the whole memory for us. Another important thing is that the far jump we used loaded 8 into CS but CS is now a selector so it means Index=1, TI=0, and RPL=0 and therefore the actual descriptor loaded is the one at index 1 in the GDT.

1.1. 32bit Programming

Our next example is to give a falvour of 32bit programming. We have written the printstr function for read and for protected mode. The availability of larger registers and flexible addressing rules allows writing a much comprehensive version of the code. Also offsets to parameters and default widths change.

	[org 0x0100]

              jmp  start

gdt:          dd   0x00000000, 0x00000000   ; null descriptor

              dd   0x0000FFFF, 0x00CF9A00   ; 32bit code

              dd   0x0000FFFF, 0x00CF9200   ; data

gdtreg:       dw   0x17                     ; 16bit limit

              dd   0                        ; 32bit base

rstring:      db   'In Real Mode...', 0

pstring:      db   'In Protected Mode...', 0

stack:        times 256 dd 0                ; 1K stack

stacktop:

printstr:     push bp                       ; real mode print string

              mov  bp, sp

              push ax

              push cx

              push si

              push di

              push es

              mov  di,[bp+4]  ;load string address

              mov  cx,0xffff  ;load maximum possible size in cx

              xor  al,al      ;clear al reg

              repne scasb     ;repeat scan

              mov ax,0xffff   ;

              sub ax,cx       ;calculate length

              dec ax          ;off by one, as it includes zero

              mov cx,ax       ;move length to counter

              mov ax, 0xb800

              mov es, ax               ; point es to video base

              mov ax,80                ;its a word move, clears ah

              mul byte [bp+8]         ;its a byte mul to calc y offset

              add ax,[bp+10]           ;add x offset

              shl ax,1                 ;mul by 2 to get word offset

              mov di,ax                ;load pointer

              mov si, [bp+4]           ; string to be printed

              mov ah, [bp+6]           ; load attribute

              cld                      ; set auto increment mode

nextchar:     lodsb                    ;load next char and inc si by 1

              stosw                    ;store ax and inc di by 2

              loop nextchar

              pop es

              pop di

              pop si

              pop cx

              pop ax

              pop bp

              ret 8

start:        push byte 0              ; 386 can directly push immediates

              push byte 10

              push byte 7

              push word rstring

              call printstr

              mov  ax, 0x2401

              int  0x15                ; enable a20

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              mov  [gdt+0x08+2], ax

              shr  eax, 16

              mov  [gdt+0x08+4], al         ; set base of code desc

              xor  edx, edx

              mov  dx, cs

              shl  edx, 4

              add  edx, stacktop       ; stacktop to be used in p-mode

              xor  ebx, ebx

              mov  bx, cs

              shl  ebx, 4

              add  ebx, pstring        ; pstring to be used in p-mode

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              add  eax, gdt

              mov  [gdtreg+2], eax          ; set base of gdt

              lgdt [gdtreg]                 ; load gdtr

              mov  eax, cr0

              or   eax, 1

              cli                           ; disable interrupts

              mov  cr0, eax                 ; enable protected mode

              jmp  0x08:pstart              ; load cs

;;;;; 32bit protected mode ;;;;;

[bits 32]

pprintstr:    push ebp                 ; p-mode print string routine

              mov  ebp, esp

              push eax

              push ecx

              push esi

              push edi

              mov  edi, [ebp+8]  ;load string address

              mov  ecx, 0xffffffff  ;load maximum possible size in cx

              xor  al, al      ;clear al reg

              repne scasb     ;repeat scan

              mov  eax, 0xffffffff   ;

              sub  eax, ecx       ;calculate length

              dec  eax          ;off by one, as it includes zero

              mov  ecx, eax       ;move length to counter

              mov  eax, 80                ;its a word move, clears ah

              mul  byte [ebp+16]         ;its a byte mul to calc y offset

              add  eax, [ebp+20]           ;add x offset

              shl  eax, 1                 ;mul by 2 to get word offset

              add  eax, 0xb8000

              mov  edi, eax                ;load pointer

              mov esi, [ebp+8]           ; string to be printed

              mov ah, [ebp+12]           ; load attribute

              cld                      ; set auto increment mode

pnextchar:    lodsb                    ;load next char and inc si by 1

              stosw                    ;store ax and inc di by 2

              loop pnextchar

              pop  edi

              pop  esi

              pop  ecx

              pop  eax

              pop  ebp

              ret  16                  ; 4 args now mean 16 bytes

pstart:       mov  ax, 0x10            ; load all seg regs to 0x10

              mov  ds, ax              ; flat memory model

              mov  es, ax

              mov  fs, ax

              mov  gs, ax

              mov  ss, ax

              mov  esp, edx            ; load saved esp on stack

              push byte 0

              push byte 11

              push byte 7

              push ebx

              call pprintstr           ; call p-mode print string routine

              mov  eax, 0x000b8000

              mov  ebx, '/-\|'

nextsymbol:   mov  [eax], bl

              mov  ecx, 0x00FFFFFF

              loop $

              ror  ebx, 8

              jmp  nextsymbol


1.2. VESA Linear Frame Buffer

As an example of accessing a really large area of memory for which protected mode is a necessity, we will be accessing the video memory in high resolution and high color graphics mode where the necessary video memory is alone above a megabyte. We will be using the VESA VBE 2.0 for a standard for these high resolution modes.

VESA is the Video Electronics Standards Association and VBE is the set of Video BIOS Extensions proposed by them. The VESA VBE 2.0 standard includes a linear frame buffer mode that we will be using. This mode allows direct access to the whole video memory. Some important VESA services are listed below.

INT 10 – VESA – Get  SuperVGA Infromation

AX = 4F00h

ES:DI -> buffer for SuperVGA information

Return: 
AL = 4Fh if function supported

AH = status

INT 10 – VESA – Get SuperVGA Mode Information

AX = 4F01h

CX = SuperVGA video mode

ES:DI -> 256-byte buffer for mode information

Return: 
AL = 4Fh if function supported

AH = status
ES:DI filled if no error
INT 10 – VESA – Set VESA Video Mode

AX = 4F02h

BX = new video mode

Return: 
AL = 4Fh if function supported

AH = status

One of the VESA defined modes is 8117 which is a 1024x768 mode with 16bit color and a linear frame buffer. The 16 color bits for every pixel are organized in 5:6:5 format with 5 bits for red, 6 for green, and 5 for blue. This makes 32 shades of red and blue and 64 shades of green and 64K total possible colors. The 32bit linear frame buffer base address is available at offset 28 in the mode information buffer. Our example will produces shades of green on the screen and clear them and again print them in an infinite loop with delays inbetween.

	[org 0x0100]

              jmp  start

modeblock:    times 256 db 0

gdt:          dd   0x00000000, 0x00000000   ; null descriptor

              dd   0x0000FFFF, 0x00CF9A00   ; 32bit code

              dd   0x0000FFFF, 0x00CF9200   ; data

gdtreg:       dw   0x17                     ; 16bit limit

              dd   0                        ; 32bit base

stack:        times 256 dd 0                ; 1K stack

stacktop:

start:        mov  ax, 0x4f01          ; get vesa mode information

              mov  cx, 0x8117          ; 1024*768*64K linear frame buffer

              mov  di, modeblock

              int  0x10

              mov  esi, [modeblock+0x28]    ; save frame buffer base
              mov  ax, 0x4f02          ; set vesa mode

              mov  bx, 0x8117

              int  0x10

              mov  ax, 0x2401

              int  0x15                ; enable a20

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              mov  [gdt+0x08+2], ax

              shr  eax, 16

              mov  [gdt+0x08+4], al         ; set base of code desc

              xor  edx, edx

              mov  dx, cs

              shl  edx, 4

              add  edx, stacktop       ; stacktop to be used in p-mode

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              add  eax, gdt

              mov  [gdtreg+2], eax          ; set base of gdt

              lgdt [gdtreg]                 ; load gdtr

              mov  eax, cr0

              or   eax, 1

              cli                           ; disable interrupts

              mov  cr0, eax                 ; enable protected mode

              jmp  0x08:pstart              ; load cs

;;;;; 32bit protected mode ;;;;;

[bits 32]

pstart:       mov  ax, 0x10            ; load all seg regs to 0x10

              mov  ds, ax              ; flat memory model

              mov  es, ax

              mov  fs, ax

              mov  gs, ax

              mov  ss, ax

              mov  esp, edx            ; load saved esp on stack

l1:           xor  eax, eax

              mov  edi, esi

              mov  ecx, 1024*768*2/4        ; divide by 4 as dwords

              cld

              rep  stosd

              mov  eax, 0x07FF07FF

              mov  ecx, 32


; no of bands

              mov  edi, esi

l2:           push ecx

              mov  ecx, 768*16


; band width = 32 lines

              cld

              rep stosd

              mov  ecx, 0x000FFFFF

; small wait

              loop $

              pop  ecx

              sub  eax, 0x00410041

              loop l2

              mov  ecx, 0x0FFFFFFF

; long wait

              loop $

              jmp  l1


1.3. Interrupt Handling

Handling interrupts in protected mode is also different. Instead of the IVT at physical address 0 there is the IDT (interrupt descriptor table) located at physical address stored in IDTR, a special purpose register. The IDTR is also a 48bit register similar in structure to the GDTR and loaded with another special instruction LGDT. The format of the interrupt descriptor is as shown below.

The P and DPL have the same meaning as in data and code descriptors. The S bit tells that this is a system descriptor while the 1110 following it tells that it is a 386 interrupt gate. Our example hooks the keyboard and timer interrupts and displays certain things on the screen to show that they are working.

	[org 0x0100]

              jmp  start

gdt:          dd   0x00000000, 0x00000000   ; null descriptor

              dd   0x0000FFFF, 0x00CF9A00   ; 32bit code

              dd   0x0000FFFF, 0x00CF9200   ; data

gdtreg:       dw   0x17                     ; 16bit limit

              dd   0                        ; 32bit base

idt:          times 8 dw unhandled, 0x0008, 0x8e00, 0x0000

              dw   timer, 0x0008, 0x8e00, 0x0000

                   \---/  \----/    ||\/  \----/

                     |      |       || |    +----- offset bits 16..32
                     |      |       || +----- reserved

                     |      |       |+------ Type=E 386 Interrupt Gate
                     |      |       +--- P=1 DPL=00 S=0

                     |      +------- selector

                     +-------- offset bits 0..15
              dw   keyboard, 0x0008, 0x8e00, 0x0000

              times 246 dw unhandled, 0x0008, 0x8e00, 0x0000

idtreg:       dw   0x07FF

              dd   0

stack:        times 256 dd 0                ; 1K stack

stacktop:

start:        mov  ax, 0x2401

              int  0x15                ; enable a20

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              mov  [gdt+0x08+2], ax

              shr  eax, 16

              mov  [gdt+0x08+4], al         ; set base of code desc

              xor  edx, edx

              mov  dx, cs

              shl  edx, 4

              add  edx, stacktop       ; stacktop to be used in p-mode

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              add  eax, gdt

              mov  [gdtreg+2], eax          ; set base of gdt

              lgdt [gdtreg]                 ; load gdtr

              xor  eax, eax

              mov  ax, cs

              shl  eax, 4

              add  eax, idt

              mov  [idtreg+2], eax          ; set base of idt

              cli                           ; disable interrupts

              lidt [idtreg]                 ; load idtr

              mov  eax, cr0

              or   eax, 1

              mov  cr0, eax                 ; enable protected mode

              jmp  0x08:pstart              ; load cs

;;;;; 32bit protected mode ;;;;;

[bits 32]

unhandled:    iret

timer:        push eax

              inc  byte [0x000b8000]


      mov  al, 0x20

              out  0x20, al

              pop  eax

              iret

keyboard:     push eax

              in   al, 0x60

              mov  ah, al

              and  al, 0x0F

              shr  ah, 4

              add  ax, 0x3030

              cmp  al, 0x39

              jbe  skip1

              add  al, 7

skip1:        cmp  ah, 0x39

              jbe  skip2

              add  ah, 7

skip2:        mov  [0x000b809C], ah

              mov  [0x000b809E], al

skipkb:       mov  al, 0x20

              out  0x20, al

              pop  eax

              iret

pstart:       mov  ax, 0x10            ; load all seg regs to 0x10

              mov  ds, ax              ; flat memory model

              mov  es, ax

              mov  fs, ax

              mov  gs, ax

              mov  ss, ax

              mov  esp, edx            ; load saved esp on stack

              mov  al, 0xFC

              out  0x21, al            ; no unexpected int comes

              sti                      ; interrupts are okay now

              jmp  $


Exercises

1. Write very brief and to-the-point answers.
a. Why loading idtr with a value appropriate for real mode is necessary while gdtr is not?

b. What should we do in protected mode so that when we turn protection off, we are in unreal mode?

c. If the line jmp code:next is replaced with call code:somefun, the prefetch queue is still emptied. What problem will occur when somefun will return?

d. How much is ESP decremented when an interrupt arrives. This depends on weather we are in 16-bit mode or 32-bit. Does it depend on any other thing as well? If yes, what?
e. Give two instructions that change the TR register.
2. Name the following descriptors like code descriptor, data descriptor, interrupt gate etc.

gdt: 
dd 0x00000000, 0x00000000
dd 0x00000000, 0x00000000
dd 0x80000fA0, 0x0000820b
dd 0x0000ffff, 0x00409a00
dd 0x80000000, 0x0001d20b

3. Using the above GDT, which of the following values, when moved into DS will cause an exception and why.

0x00
0x08
0x10
0x18
0x28
0x23

4. Using the above GDT, if DS contains 0x20, which of the following offsets will cause an exception on read access?

0x0ffff
0x10000
0x10001

5. The following function is written in 32-bit code for a 16-bit stack. Against every instruction, write the prefixes generated before that instruction. Prefixes can be address size, operand size, repeat, or segment override. Then rewrite the code such that no prefixes are generated considering that this is assembled and executed in 32-bit mode. Don’t care for retaining register values. The function copies specified number of DWORDs between two segments.
[bits 32]
memcpy:       mov  bp, sp             
              lds  esi, [bp+4]        ; source address
              les  edi, [bp+10]       ; destination address
              mov  cx, [bp+16]        ; count of DWORDs to move
              shl  cx, 1              ; make into count of WORDs
L1:           mov  dx, [si]
              mov  [es:di], dx
              dec  cx
              jnz  L1
              ret

6. Rewrite the following scheduler so that it schedules processes stored in readyQ, where enque and deque functions are redefined and readyQ contains TSS selectors of processes to be multitasked. Remember you can’t use a register as a segment in a jump (eg jmp ax:0) but you can jump to an indirect address (eg jmp far [eax]) where eax points to a six-byte address. Declare any variables you need.

              mov al, 0x20
scheduler:    jmp USERONESEL:0
              out  0x20, al
              mov  byte [USERONEDESC+5], 0x89
              jmp  USERTWOSEL:0
              out  0x20, al
              mov  byte [USERTWODESC+5], 0x89
              jmp  scheduler

7. Protected mode has specialized mechanism for multitasking using task state segments but the method used in real mode i.e. saving all registers in a PCB, selecting the next PCB and loading all registers from there is still applicable. Multitask two tasks in protected mode multitasking without TSS. Assume that all processes are at level zero so no protection issues arise. Be careful to save the complete state of the process.

8. Write the following descriptors. 

a. 32 bit, conforming, execute-only code segment at level 2, with base at 6MB and a size of 4MB.

b. 16 bit, non-conforming, readable code segment at level 0, with base at 1MB and a size of 10 bytes.

c. Read only data segment at level 3, with base at 0 and size of 1MB.

d. Interrupt Gate with selector 180h and offset 11223344h.

9. Write physical addresses for the following accesses where CS points to the first descriptor above, DS to the second, ES to the third, EBX contains 00010000h, and ESI contains 00020000h

a. [bx+si]
b. [ebx+esi-2ffffh]
c. [es:ebx-10h]
10. Which of the following will cause exceptions and why. The registers have the same values as the last question.

a. mov eax, [cs:10000h]
b. mov [es:esi:100h], ebx
c. mov ax, [es:ebx]
11. Give short answers.
a. How can a GPF (General protection fault) occur while running the following code

push es
pop  es
b. How can a GPF occur during the following instruction? Give any two reasons.

jmp  10h:100h

c. What will happen if we call interrupt 80h after loading out IDT and before switching to protected mode?

d. What will happen if we call interrupt 80h after switching into protected mode but before making a far jump?

12. Write the following descriptors. Assume values for attributes not specifically mentioned.
a. Write able 32-bit data segment with 1 GB base and 1 GB limit and a privilege level of 2.
b. Readable 16-bit code descriptor with 1 MB base and 1 MB limit and a privilege level of 1.
c. Interrupt gate given that the handler is at 48h:12345678h and a privilege level of 0.
13. Describe the following descriptors. Give their type and the value of all their fields.
dd 01234567h, 789abcdeh
dd 30405060h, 70809010h
dd 00aabb00h, 00ffee00h

14. Make an EXE file, switch into protected mode, rotate an asterisk on the border of the screen, and return to real mode when the border is traversed.
